Komputeralgebrai algoritmusok

Járai Antal

Ezek a programok csak szemléltetésre szolgálnak.

1. Történet

2. Algebrai alapok

3. Normál formák, reprezentáció

4. Aritmetika

5. Kínai maradékolás

6. Newton-iteráció, Hensel-felemelés

7. Legnagyobb közös osztó

8. Faktorizálás

9. Egyenletrendszerek

10. Gröbner-bázisok

E 10.1. Lexikografikus rendezés.

```
> p:=x+y+z+x^2+z^3+x*y+y^2;
```

$$p := x + y + z + x^2 + z^3 + x\,y + y^2 \tag{10.1.1}$$

```
> sort(p,[x,y,z],plex);
```

$$x^2 + x\,y + x + y^2 + y + z^3 + z \tag{10.1.2}$$

```
> sort(p,[z,y,x],plex);
```

$$z^3 + z + y^2 + y\,x + y + x^2 + x \tag{10.1.3}$$

E 10.2. Fokszám rendezés.

```
> sort(p,[x,y,z]);
```

$$z^3+x^2+x\,y+y^2+x+y+z \quad (10.2.1)$$

E 10.3. Példa.

```
> p:=-2*x^2*y*z+x^2*y^2+x^2*z^2+x^2*y+2*x*y^2*z^2-3*x*y*z^3-x*
  y+y*z+z^2+5;
```

$$p := -2\,x^2\,y\,z+x^2\,y^2+x^2\,z^2+x^2\,y+2\,x\,y^2\,z^2-3\,x\,y\,z^3-x\,y+y\,z+z^2+5 \quad (10.3.1)$$

```
> sort(p,[x,y,z]);
```

$$2\,x\,y^2\,z^2-3\,x\,y\,z^3+x^2\,y^2-2\,x^2\,y\,z+x^2\,z^2+x^2\,y-x\,y+y\,z+z^2+5 \quad (10.3.2)$$

```
> op(1,p); lcoeff(%,[x,y,z],'t'); t;
```

$$2\,x\,y^2\,z^2$$
$$2$$
$$x\,y^2\,z^2 \quad (10.3.3)$$

```
> collect(p,[x,y]);
```

$$\left(y^2+\left(-2\,z+1\right)y+z^2\right)x^2+\left(2\,y^2\,z^2+\left(-1-3\,z^3\right)y\right)x+z^2+5+y\,z \quad (10.3.4)$$

```
> pp:=collect(p,[x,y],`distributed`);
```

$$pp := z^2+5+2\,x\,y^2\,z^2+\left(-1-3\,z^3\right)x\,y+x^2\,y^2+\left(-2\,z+1\right)x^2\,y+y\,z+x^2\,z^2 \quad (10.3.5)$$

```
> sort(pp,[x,y]);
```

$$x^2\,y^2+\left(-2\,z+1\right)x^2\,y+2\,z^2\,x\,y^2+z^2\,x^2+\left(-1-3\,z^3\right)x\,y+z\,y+z^2+5 \quad (10.3.6)$$

```
> op(1,p); lcoeff(%,[x,y],'t'); t;
```

$$2\,z^2\,x\,y^2$$
$$2\,z^2$$
$$y^2\,x \quad (10.3.7)$$

```
> with(Groebner);
```

[*Basis*, *FGLM*, *HilbertDimension*, *HilbertPolynomial*, *HilbertSeries*, *InterReduce*, *IsProper*, *IsZeroDimensional*, *LeadingCoefficient*, *LeadingMonomial*, *LeadingTerm*, *MonomialOrder*, *MultiplicationMatrix*, *NormalForm*, *NormalSet*, *Reduce*, *RememberBasis*, *SPolynomial*, *Solve*, *TestOrder*, *ToricIdealBasis*, *UnivariatePolynomial*, *Walk*, *fglm_algo*] (10.3.8)

```
> LeadingTerm(p,tdeg(x,y,z)); %[1]*%[2];
```

$$2,\ x\,y^2\,z^2$$
$$2\,x\,y^2\,z^2 \quad (10.3.9)$$

```
> LeadingMonomial(p,tdeg(x,y,z));
```

$$x\,y^2\,z^2 \quad (10.3.10)$$

```
> LeadingCoefficient(p,tdeg(x,y,z));
```
$$2 \tag{10.3.11}$$

E 10.4. Példa.

```
> p:=6*x^4+13*x^3-6*x+1; q:=3*x^2+5*x-1;
```
$$p := 6\,x^4 + 13\,x^3 - 6\,x + 1$$
$$q := 3\,x^2 + 5\,x - 1 \tag{10.4.1}$$

```
> p-2*x^2*q;
```
$$6\,x^4 + 13\,x^3 - 6\,x + 1 - 2\,x^2\left(3\,x^2 + 5\,x - 1\right) \tag{10.4.2}$$

```
> expand(p-2*x^2*q);
```
$$3\,x^3 - 6\,x + 1 + 2\,x^2 \tag{10.4.3}$$

```
> expand(p-13/3*x*q);
```
$$6\,x^4 - \frac{5}{3}\,x + 1 - \frac{65}{3}\,x^2 \tag{10.4.4}$$

E 10.5. Példa.

```
> p:=2*y^2*z-x*z^2; q1:=7*y^2+y*z-4; q2:=2*y*z-3*x+1;
```
$$p := 2\,y^2\,z - x\,z^2$$
$$q1 := 7\,y^2 + y\,z - 4$$
$$q2 := 2\,y\,z - 3\,x + 1 \tag{10.5.1}$$

```
> sort(p,[x,y,z]); sort(q1,[x,y,z]); sort(q2,[x,y,z]);
```
$$-x\,z^2 + 2\,y^2\,z$$
$$7\,y^2 + y\,z - 4$$
$$2\,y\,z - 3\,x + 1 \tag{10.5.2}$$

```
> expand(p-2/7*z*q1); expand(p-y*q2);
```
$$-x\,z^2 - \frac{2}{7}\,y\,z^2 + \frac{8}{7}\,z$$
$$-x\,z^2 + 3\,x\,y - y \tag{10.5.3}$$

```
> expand(%+1/7*q2);
```
$$-x\,z^2 + 3\,x\,y - y + \frac{2}{7}\,y\,z - \frac{3}{7}\,x + \frac{1}{7} \tag{10.5.4}$$

A 10.1. Algoritmus.

```
> reducers:=proc(p,Q,ord) local phmonom,qhmonom,q,R;
```

```
    R:=[]; phmonom:=LeadingMonomial(p,ord);
    for q in Q do
      if q<>0 then
        qhmonom:=LeadingMonomial(q,ord);
        if divide(phmonom,qhmonom) then R:=[op(R),q]; fi;
      fi;
    od; R;
  end;
```

reducers := **proc**(*p*, *Q*, *ord*) (10.6.1)

local *phmonom*, *qhmonom*, *q*, *R*;

R := [ ];

phmonom := *Groebner*:-*LeadingMonomial*(*p*, *ord*);

for *q* **in** *Q* **do**

if $q <> 0$ **then**

qhmonom := *Groebner*:-*LeadingMonomial*(*q*, *ord*);

if *divide*(*phmonom*, *qhmonom*) **then**

R := [*op*(*R*), *q*]

end if

end if

end do;

R

end proc

```
> Q:=[q1,q2]; QQ:=reducers(p,Q,tdeg(x,y,z));
```

$$Q := [7y^2 + yz - 4, 2yz - 3x + 1]$$

$$QQ := [7y^2 + yz - 4, 2yz - 3x + 1] \qquad (10.6.2)$$

```
> reducer:=proc(p,Q,ord) Q[1] end;
```

reducer := **proc**(*p*, *Q*, *ord*) *Q*[1] **end proc** (10.6.3)

```
> reducer(p,QQ,tdeg(x,y,z));
```

$$7y^2 + yz - 4 \qquad (10.6.4)$$

```
> myreduce:=proc(p,Q,ord) local QQ,q,pp,ppp,lpp,lq,t;
    pp:=p; ppp:=0;
    while pp<>0 do
      QQ:=reducers(pp,Q,ord);
      if QQ<>[] then
        q:=reducer(p,QQ,ord);
        t:=LeadingTerm(pp,ord); lpp:=t[1]*t[2];
        t:=LeadingTerm(q,ord); lq:=t[1]*t[2];
        pp:=expand(pp-lpp*q/lq);
      else
        LeadingTerm(pp,ord); lpp:=%[1]*%[2];
        ppp:=ppp+lpp;
```

```
            pp:=pp-lpp;
          fi;
        od; ppp;
    end;
```

$myreduce := \mathbf{proc}(p, Q, ord)$ (10.6.5)

```
    local QQ, q, pp, ppp, lpp, lq, t;
    pp := p;
    ppp := 0;
    while pp <> 0 do
        QQ := reducers(pp, Q, ord);
        if QQ <> [ ] then
            q := reducer(p, QQ, ord);
            t := Groebner:-LeadingTerm(pp, ord);
            lpp := t[1] * t[2];
            t := Groebner:-LeadingTerm(q, ord);
            lq := t[1] * t[2];
            pp := expand(pp - lpp * q / lq)
        else
            Groebner:-LeadingTerm(pp, ord);
            lpp := %[1] * %[2];
            ppp := ppp + lpp;
            pp := pp - lpp
        end if
    end do;
    ppp
end proc
```

```
> myreduce(p,Q,tdeg(x,y,z));
```

$$-x z^2 - \frac{3}{7} z x + \frac{9}{7} z \quad (10.6.6)$$

E 10.6. Példa.

```
> p:=3*x^3*y+2*x^2*y^2-3*x*y+5*x; q1:=x^2*y+5*x^2+y^2; q2:=7*x*
  y^2-2*y^3+1;
```

$$p := 3 x^3 y + 2 x^2 y^2 - 3 x y + 5 x$$

$$q1 := x^2 y + 5 x^2 + y^2$$

$$q2 := 7 x y^2 - 2 y^3 + 1 \quad (10.7.1)$$

```
> expand(p-3*x*q1);
```

(10.7.2)

$$2x^2y^2 - 3xy + 5x - 15x^3 - 3xy^2 \quad (10.7.2)$$

```
> expand(%-2*y*q1);
```

$$-3xy + 5x - 15x^3 - 3xy^2 - 10x^2y - 2y^3 \quad (10.7.3)$$

```
> expand(%+10*q1);
```

$$-3xy + 5x - 15x^3 - 3xy^2 - 2y^3 + 50x^2 + 10y^2 \quad (10.7.4)$$

```
> expand(%+3/7*q2);
```

$$-3xy + 5x - 15x^3 - \frac{20}{7}y^3 + 50x^2 + 10y^2 + \frac{3}{7} \quad (10.7.5)$$

E 10.7. Példa.

```
> q1:=x^3*y*z-x*z^2; q2:=x*y^2*z-x*y*z; q3:=x^2*y^2-z^2;
  p1:=x^2*y^2*z-z^3; p2:=-x^2*y^2*z+x^2*y*z;
```

$$q1 := x^3yz - xz^2$$

$$q2 := xy^2z - xyz$$

$$q3 := x^2y^2 - z^2$$

$$p1 := x^2y^2z - z^3$$

$$p2 := -x^2y^2z + x^2yz \quad (10.8.1)$$

```
> expand(p1-z*q3); expand(p2+x*q2); p1+p2;
```

$$0$$

$$0$$

$$-z^3 + x^2yz \quad (10.8.2)$$

E 10.8. Példa.

```
> G:=[q1,q2,q3,x^2*y*z-z^3,x*z^3-x*z^2,y*z^3-z^3,x*y*z^2-x*z^2,
  x^2*z^2-z^4,
  z^5-z^4];
```

$$G := [x^3yz - xz^2, xy^2z - xyz, x^2y^2 - z^2, -z^3 + x^2yz, xz^3 - xz^2, yz^3 - z^3, xyz^2 - xz^2, x^2z^2 - z^4, z^5 - z^4] \quad (10.9.1)$$

```
> myreduce(p1,G,tdeg(x,y,z)); myreduce(p2,G,tdeg(x,y,z));
  myreduce(p1-p2,G,tdeg(x,y,z));
```

$$0$$

$$0$$

$$0 \quad (10.9.2)$$

▶ **12. A Risch–algoritmus.**

▶ **13. ő**